ENTERPRISE
COMPUTERS LTD

31-37 Hoxton Street
London N1 6NJ
Telephone 01-739 4282
Telex 22717 ENTER G

APPLICATION NOTE NO. 13

DUMPING GRAPHICS TO THE PRINTER

The 'dumping' of graphics screens to a printer requires a short BASIC program. A program to do this is included in the form of a procedure.

The computer scans the screen 8 bits at a time, and puts the data into a 180 character buffer. This is then sent to the printer.

Because of the way the graphics screen is encoded, this program will only work in the low resolution graphics modes. However, as LORES 2 has higher resolution than most other computers can manage, this should not be a problem.

To 'dump' an ordinary text screen a program is not required, just type COPY and press ENTER. Here is a sample program to draw a picture and 'dump' it to the printer. The control codes used are for the EP80 and printer, but will work on any Mannesmann Tally printer and most EPSON compatible machines.

GM/LLB/5/6/85.

```
100 PROGRAM "GDUMP_2"
110 GRAPHICS LORES 2
120 SET PALETTE 7,255
130 FOR X=10 TO 100 STEP 12
140   FOR Y=1 TO 1160 STEP 90
150     PLOT Y+X,120+100*SIN(Y/70*PI)+X;
160   NEXT
170   SET BEAM OFF
180 NEXT
190 PLOT 630,360,;ELLIPSE 120,120,PAINT
200 PLOT 30,680,;ELLIPSE 270,270,PAINT
210 PLOT 900,570,
220 FOR X=1 TO 200 STEP 12
230   PLOT ELLIPSE X,200-X,
240 NEXT
250 PLOT 20,400,
260 SET LINE MODE 3
270 PRINT £101:"      Written by Gerald Morgan."
280 !
290 CALL GDUMP
300 !
310 DEF GDUMP
320   NUMERIC DOT(180)
330   LET E$=CHR$(27):LET START=36146:LET FINISH=START+32
340   LPRINT E$;"@";
350   LPRINT E$;"3";CHR$(21);
360   FOR LOOP1=START TO FINISH
370     LPRINT E$;"K";CHR$(180);CHR$(0);
380     LET COUNT=0
390     FOR LOOP2=LOOP1 TO 7199+LOOP1 STEP 40
400       LET DOT(COUNT)=PEEK(LOOP2)
410       LET COUNT=COUNT+1
420     NEXT
430     FOR LOOP3=179 TO 0 STEP-1
440       LPRINT CHR$(DOT(LOOP3));
450     NEXT
460     LPRINT
470   NEXT
480 END DEF
```